
Projet Collaboratif

Problème de Stokes
à frontière libre

ROXANA DIA
IBRAHIMA DIAO
MAMADOU CAMARA

Encadrés par Mr CORNEL MUREA

29 décembre 2015

Université de Strasbourg



$UDS - UPMC$

MASTER II CSMI

Table des matières

1	Introduction	3
2	Formulation variationnelle	4
3	Discrétisation	6
4	Résultats	7
4.1	Premier cas	7
4.2	Deuxième cas	11
5	Code	15

1 Introduction

Dans ce projet, on se donne un domaine $\Omega \in \mathbb{R}^2$ fixe divisé en deux sous-domaines Ω_t^F et Ω_t^G qui varient au cours du temps. Ω_t^F représente le domaine du fluide et Ω_t^G le domaine du gaz. La frontière du domaine Ω est constituée par :

$$\Gamma_1 = \{(x, y) \ ; \ x = 0 \quad \text{et} \quad -H_2 \leq y \leq H_1\}$$

$$\Gamma_2 = \{(x, y) \ ; \ y = H_1 \quad \text{et} \quad 0 \leq x \leq R_1\}$$

$$\Gamma_3 = \{(x, y) \ ; \ x = R_1 \quad \text{et} \quad 0 \leq y \leq H_1\}$$

$$\Gamma_4 = \{(x, y) \ ; \ y = 0 \quad \text{et} \quad R_1 \leq x \leq R_2\}$$

$$\Gamma_5 = \{(x, y) \ ; \ x^2 + y^2 = R_2^2 \quad \text{et} \quad x \geq 0 \quad \text{et} \quad y \leq 0\}$$

Où $R_1 = 10 \ \mu m$, $R_2 = 40 \ \mu m$, $H_1 = 40 \ \mu m$ et $H_2 = 60 \ \mu m$.

D'autre part, dans Ω on retrouve Γ_t qui désigne l'interface entre Ω_t^F et Ω_t^G . Γ_t est définie à partir d'une fonction ϕ telle que $\phi(x, y, t) = 0$ sur Γ_t , $\phi(x, y, t) > 0$ sur Ω_t^F et $\phi(x, y, t) < 0$ sur Ω_t^G . On teste deux formes différentes de cette fonction dans la suite.

On se donne les équations de Stokes dans Ω_t^F et Ω_t^G définies comme suit :

$$\begin{cases} \rho^F \frac{\partial u^F}{\partial t} - \nabla \cdot \sigma^F = f^F & \Omega_t^F \\ \nabla \cdot u^F = 0 & \Omega_t^F \end{cases} \quad (1)$$

Où,

$u^F = (u_1^F, u_2^F)$ représente la vitesse du fluide.

$\sigma^F = 2\mu^F \varepsilon(u^F) - p^F I$; p^F étant la pression du fluide.

$\varepsilon(u^F) = \frac{1}{2}(\nabla u^F + (\nabla u^F)^T)$.

f^F est la force appliquée au fluide.

$\mu^F = 0.00334 \frac{mg}{\mu m.s}$ est la viscosité dynamique du fluide.

$\rho^F = 1070 * 10^{-12} \frac{mg}{\mu m^3}$

De la même manière pour le gaz, on cherche une vitesse $u^G = (u_1^G, u_2^G)$ et une pression p^G vérifiant le système suivant :

$$\begin{cases} \rho^G \frac{\partial u^G}{\partial t} - \nabla \cdot \sigma^G = f^G & \Omega_t^G \\ \nabla \cdot u^G = 0 & \Omega_t^G \end{cases} \quad (2)$$

Dans ce cas les coefficients sont ceux du gaz, on garde les mêmes définitions que dans (1), avec :

$\mu^G = 0.00177625 \frac{mg}{\mu m.s}$

$\rho^G = 1.225 * 10^{-12} \frac{mg}{\mu m^3}$.

Notons que dans notre cas $f^F = f^G = 0$.

D'autre part, on se donne les conditions aux bords suivantes :

Sur Γ_1 : $u \cdot n = 0$, $(\sigma n) \cdot \tau = 0$.

Sur Γ_2 : $u_1 = 0$, $u_2 = \frac{v_{in}(x-R_1)(x+R_1)}{R_1^2}$, avec $v_{in} = 10 \mu m.s^{-1}$.

Sur $\Gamma_3 \cup \Gamma_4$: $u_1 = 0$, $u_2 = 0$.

Sur Γ_5 : $\sigma n = h = 0$.

Sur Γ_t : $u^G = u^F$ et $\sigma^F n^F + \sigma^G n^G = -\gamma K n^F$ ($\gamma = 10^{-6} mg.s^{-2}$, K est la courbure de Γ_t et n^F la normale extérieure au domaine Ω_t^F sur Γ_t).

Où u , σ et p définis comme suit,

$$u = \begin{cases} u^F & \text{dans } \Omega_t^F \\ u^G & \text{dans } \Omega_t^G \end{cases}$$

$$\sigma = \begin{cases} \sigma^F & \text{dans } \Omega_t^F \\ \sigma^G & \text{dans } \Omega_t^G \end{cases}$$

$$p = \begin{cases} p^F & \text{dans } \Omega_t^F \\ p^G & \text{dans } \Omega_t^G \end{cases}$$

2 Formulation variationnelle

Notons tout d'abord que Ω_t^F est limité par Γ_2 , Γ_3 , Γ_t et une partie de Γ_1 . D'autre part Ω_t^G est limité par Γ_4 , Γ_5 , Γ_t et une partie de Γ_1 .

On commence par chercher la formulation variationnelle dans Ω_t^F . On multiplie la première équation du problème (1) par une fonction test $v^F \in (H^1(\Omega_t^F))^2$ telle que $v^F = 0$ sur $\Gamma_2 \cup \Gamma_3$ et $v^F \cdot n^F = 0$ sur Γ_1 (ie : $v^F = 0$ sur Γ_1 car $n^F = (-1, 0)$).

On obtient alors :

$$\rho^F \int_{\Omega_t^F} \frac{\partial u^F}{\partial t} \cdot v^F dx - \int_{\Omega_t^F} (\nabla \cdot \sigma^F) \cdot v^F = \int_{\Omega_t^F} f^F \cdot v^F$$

Or pour tout u , $v \in (H^1(\Omega))^2$ Où Ω étant un domaine régulier et pour $\sigma = 2\mu\varepsilon(u) - pI$, on a la formule suivante :

$$\begin{aligned} - \int_{\Omega} (\nabla \cdot \sigma) \cdot v dx &= \int_{\Omega} \sigma : \nabla v dx - \int_{\partial\Omega} \sigma n \cdot v ds \\ &= 2\mu \int_{\Omega} \varepsilon(u) : \varepsilon(v) dx - \int_{\Omega} (\nabla \cdot v) p dx - \int_{\partial\Omega} \sigma n \cdot v ds \end{aligned}$$

La formulation variationnelle s'écrit alors :

$$\rho^F \int_{\Omega_t^F} \frac{\partial u^F}{\partial t} \cdot v^F dx + 2\mu \int_{\Omega_t^F} \varepsilon(u^F) : \varepsilon(v^F) dx - \int_{\Omega_t^F} (\nabla \cdot v^F) p^F dx = \int_{\Omega_t^F} f^F \cdot v^F + \int_{\partial\Omega_t^F} \sigma^F n^F \cdot v^F ds$$

Or

$$\int_{\partial\Omega_t^F} \sigma^F n^F \cdot v^F ds = \int_{\Gamma_1} \sigma^F n^F \cdot v^F ds + \int_{\Gamma_2 \cup \Gamma_3} \sigma^F n^F \cdot v^F ds + \int_{\Gamma_t} \sigma^F n^F \cdot v^F ds$$

Mais d'un part $v^F = 0$ sur $\Gamma_2 \cup \Gamma_3$ et d'autre part

$$\begin{aligned} \sigma^F n^F \cdot v^F &= \sigma^F n^F \cdot (v^F \cdot n^F) n^F + \sigma^F n^F (v^F \cdot \tau^F) \tau^F \\ &= (v^F \cdot n^F) \sigma^F n^F \cdot n^F + (v^F \cdot \tau^F) (\sigma^F n^F) \cdot \tau^F \end{aligned}$$

Or sur Γ_1 $v^F \cdot n^F = 0$ et $\sigma^F n^F \cdot \tau^F = 0$

Donc

$$\int_{\partial\Omega_t^F} \sigma^F n^F \cdot v^F ds = \int_{\Gamma_t} \sigma^F n^F \cdot v^F ds$$

On obtient finalement :

$$\rho^F \int_{\Omega_t^F} \frac{\partial u^F}{\partial t} \cdot v^F dx + 2\mu \int_{\Omega_t^F} \varepsilon(u^F) : \varepsilon(v^F) dx - \int_{\Omega_t^F} (\nabla \cdot v^F) p^F dx = \int_{\Omega_t^F} f^F \cdot v^F + \int_{\Gamma_t} \sigma^F n^F \cdot v^F ds \quad (3)$$

Pareillement on multiplie la première équation de (2) par $v^G \in (H^1(\Omega_t^G))^2$ et telle que $v^G = 0$ sur Γ_4 et $v^G \cdot n^G = 0$ sur Γ_1 ,

On obtient alors :

$$\rho^G \int_{\Omega_t^G} \frac{\partial u^G}{\partial t} \cdot v^G dx + 2\mu \int_{\Omega_t^G} \varepsilon(u^G) : \varepsilon(v^G) dx - \int_{\Omega_t^G} (\nabla \cdot v^G) p^G dx = \int_{\Omega_t^G} f^G \cdot v^G + \int_{\partial\Omega_t^G} \sigma^G n^G \cdot v^G ds$$

Or $\sigma^G n^G = 0$ sur Γ_5 , $v^G = 0$ sur Γ_4 , et $v^G \cdot n^G = 0$ et $\sigma^G n^G \cdot \tau^G = 0$ sur Γ_1 .

Ainsi la formulation s'écrit :

$$\rho^G \int_{\Omega_t^G} \frac{\partial u^G}{\partial t} \cdot v^G dx + 2\mu \int_{\Omega_t^G} \varepsilon(u^G) : \varepsilon(v^G) dx - \int_{\Omega_t^G} (\nabla \cdot v^G) p^G dx = \int_{\Omega_t^G} f^G \cdot v^G + \int_{\Gamma_t} \sigma^G n^G \cdot v^G ds \quad (4)$$

En sommant (3) et (4) et en remarquant que :

$$\begin{aligned} u &= \chi^F u^F + \chi^G u^G \\ p &= \chi^F p^F + \chi^G p^G \\ \sigma &= \chi^F \sigma^F + \chi^G \sigma^G \\ \sigma^F n^F + \sigma^G n^G &= -\gamma K n^F \end{aligned}$$

La problème revient à :

Trouver $u \in C^1([0, T], (L^2(\Omega))^2) \cap C^0([0, T], (H^1(\Omega))^2)$ et $p \in C^0([0, T], L_0^2(\Omega))$

vérifiant :

Sur Γ_2 : $u_1 = 0$ et $u_2 = \frac{v_{in}(x-R_1)(x+R_1)}{R_1^2}$

Sur $\Gamma_3 \cup \Gamma_4$: $u_1 = 0$, $u_2 = 0$.

Sur Γ_t : $u^G = u^F$

et telles que :

$$\int_{\Omega} (\rho^F \chi^F + \rho^G \chi^G) \frac{\partial u}{\partial t} \cdot v dx + \int_{\Omega} (2\mu^F \chi^F + 2\mu^G \chi^G) \varepsilon(u) : \varepsilon(v) dx - \int_{\Omega} (\nabla \cdot v) p dx = \int_{\Omega} (\chi^F f^F + \chi^G f^G) \cdot v dx - \int_{\Gamma_t} \gamma K n^F \cdot v ds$$

$\forall v \in (H^1(\Omega))^2$ telle que $v = 0$ sur $\Gamma_2 \cup \Gamma_3 \cup \Gamma_4$ et $v \cdot n = 0$ sur Γ_1 et $\forall q \in L_0^2(\Omega)$

De la même manière en multipliant les deuxièmes équations des systèmes (1) et (2) par $q \in L_0^2(\Omega)$, en les sommant et en intégrant respectivement sur Ω_t^F et Ω_t^G , on obtient la seconde équation de cette formulation :

$$\int_{\Omega} (\nabla \cdot u) q = 0$$

3 Discrétisation

On divise l'intervalle de temps $[0, T]$ en N parties $[t_0, t_1, \dots, t_N = T]$ où $t_{n+1} - t_n = \Delta t$, $\forall n \in [0, N-1]$.

En faisant les approximations suivantes :

$$u^n(x, y) \simeq u(x, y, t_n)$$

$$p^n(x, y) \simeq p(x, y, t_n)$$

$$\Gamma_n \simeq \Gamma_{t_n}$$

$$\phi^n(x, y) \simeq \phi(x, y, t_n)$$

$$f_n^F(x, y) = f^F(x, y, t_n)$$

$$f_n^G(x, y) = f^G(x, y, t_n)$$

χ_n^F et χ_n^G les fonctions caractéristique de $\Omega_{t_n}^F$ et $\Omega_{t_n}^G$

La formulation variationnelle devient alors :

Pour u^n, p^n et ϕ^n donnés, trouver $u^{n+1} \in (H^1(\Omega))^2$, $p^{n+1} \in L_0^2(\Omega)$ vérifiant :

Sur Γ_2 : $u_1^{n+1} = 0$ et $u_2^{n+1} = \frac{v_{in}(x-R_1)(x+R_1)}{R_1^2}$

Sur $\Gamma_3 \cup \Gamma_4$: $u_1^{n+1} = 0$, $u_2^{n+1} = 0$.

Sur Γ_t : $u^{n+1, G} = u^{n+1, F}$

et telles que :

$$\begin{aligned} & \int_{\Omega} (\rho^F \chi_n^F + \rho^G \chi_n^G) \frac{u^{n+1} - u^n}{\Delta t} \cdot v dx + \int_{\Omega} (2\mu^F \chi_n^F + 2\mu^G \chi_n^G) \varepsilon(u^{n+1}) : \varepsilon(v) dx - \int_{\Omega} (\nabla \cdot v) p^{n+1} dx \\ & - \int_{\Omega} (\nabla \cdot u^{n+1}) q dx = \int_{\Omega} (\chi_n^F f_{n+1}^F + \chi_n^G f_{n+1}^G) \cdot v dx - \gamma \int_{\Gamma_n} K^n n^F \cdot v ds \end{aligned}$$

$\forall v \in (H^1(\Omega))^2$ telle que $v = 0$ sur $\Gamma_2 \cup \Gamma_3 \cup \Gamma_4$ et $v \cdot n$ sur Γ_1 et $\forall q \in L_0^2(\Omega)$.
Avec,

$$K^n = \frac{-(\phi_y^n)^2 \phi_{xx}^n + 2\phi_x^n \phi_y^n \phi_{xy}^n - (\phi_x^n)^2 \phi_{yy}^n}{((\phi_x^n)^2 + (\phi_y^n)^2)^{\frac{3}{2}}}$$

et

$$n^F = -\frac{\nabla \phi^n}{\|\nabla \phi^n\|}$$

Finalement, une fois on a u^{n+1} et p^{n+1} , On cherche ϕ^{n+1} par une formulation variationnelle telle que :

$$\int_{\Omega} (\phi^{n+1} - \phi^n + \Delta t u^{n+1} \cdot \nabla \phi^{n+1}) (\eta + \delta u^{n+1} \cdot \nabla \eta) dx = 0$$

$\forall \eta \in H^1(\Omega)$ et $\delta = \Delta t$ et telle que $\phi^{n+1} = \phi^n$ sur Γ_2 . Et On met à jour χ_{n+1}^F et χ_{n+1}^G et on reprend le problème et ainsi de suite.

4 Résultats

On teste ce problème pour $N = 30$ et pour deux choix différents de ϕ^0 .

4.1 Premier cas

On introduit deux variables D_0 et L_0 telles que $D_0 = 30\mu m$ et $L_0 = \frac{2}{3}H_2$. On introduit les fonctions $sign(x)$ et $ssqrt(x)$ définies de \mathbb{R} dans \mathbb{R} et $F(x, y)$ et $\phi^0(x, y)$ de \mathbb{R}^2 dans \mathbb{R} et prenant les valeurs suivantes :

$$sign(x) = \begin{cases} -1 & \text{si } x < 0 \\ 1 & \text{si } x \geq 0 \end{cases}$$

$$ssqrt(x) = sign(x) \sqrt{|x|} \quad \text{pour tout } x \in \mathbb{R}$$

$$F(x, y) = ssqrt(1 - y^2) \cdot (R_1 + D_0 y^2) - x \quad \text{pour tout } (x, y) \in \mathbb{R}^2$$

$$\phi^0(x, y) = F(x, \min(0, \frac{y}{L_0})) + \max(0, y) \quad \text{pour tout } (x, y) \in \mathbb{R}^2$$

Pour ce choix de ϕ^0 , les résultats obtenus sont les suivants.

Pour $\Delta t = 10^{-3}$ et $\delta = \Delta t$




FIGURE 1 – La vitesse du fluide-gaz à $t=11 \cdot \Delta t$ (à gauche) et à $t= 30 \cdot \Delta t$ (à droite)




FIGURE 2 – Le domaine Ω à $t=0$ (à gauche), $t=11*\Delta t$ (au milieu) et $t= 30 * \Delta t$ (à droite)

Or pour ce pas de temps, on constate qu'on ne peut pas voir le changement de Γ_t puisqu'elle varie très légèrement, donc on essaie d'augmenter le pas de temps et on obtient alors :

Pour $\Delta t = 10^{-1}$ et $\delta = \Delta t$




FIGURE 3 – La vitesse du fluide-gaz à $t=11 \cdot \Delta t$ (à gauche) et à $t= 30 \cdot \Delta t$ (à droite)




FIGURE 4 – Le domaine Ω à $t=0$ (à gauche), $t=11*\Delta t$ (au milieu) et $t= 30 * \Delta t$ (à droite)

4.2 Deuxième cas

Dans cette partie, on définit ϕ^0 par :

$$\phi^0(x, y) = -\alpha(x^2 + (y - 50)^2 - 1600) \quad \text{pour tout } (x, y) \in \mathbb{R}^2$$

Avec $\alpha = 0.01$.

Alors, pour ce choix de ϕ^0 , les résultats obtenus sont les suivants.

Pour $\Delta t = 10^{-3}$ et $\delta = \Delta t$




FIGURE 5 – La vitesse du fluide-gaz à $t=11 \cdot \Delta t$ (à gauche) et à $t= 30 \cdot \Delta t$ (à droite)




FIGURE 6 – Le domaine Ω à $t=0$ (à gauche), $t=11 \cdot \Delta t$ (au milieu) et $t=30 \cdot \Delta t$ (à droite)

Pour $\Delta t = 10^{-1}$ et $\delta = \Delta t$




FIGURE 7 – La vitesse du fluide-gaz à $t=11 \cdot \Delta t$ (à gauche) et à $t=30 \cdot \Delta t$ (à droite)




FIGURE 8 – Le domaine Ω à $t=0$ (à gauche), $t=11*\Delta t$ (au milieu) et $t= 30 * \Delta t$ (à droite)

5 Code

```

real hh = 1;
real R1 = 10; // 1e-6 m
real R2 = 40;
real H1 = 40;
real H2 = 60;
real L0=H2*2./3.;// longueur de la goutte
real D0=30; // epaisseur ???
real Lg1= H1+H2;
real Lg2= R1;
real Lg3= H1;
real Lg4 = R2-R1;
real Lg5 = (R1+R2)*pi/8;

```

```

border B1(t=0,1) { x = 0; y = -H2 + t*(H1+H2); label = 1;}
border B2(t=0,1) { x = t*R1; y = H1; label = 2;}
border B3(t=0,1) { x = R1; y = H1-t*H1; label = 3;}
border B4(t=0,1) { x = R1 + (R2-R1)*t; y = 0; label = 4;}
border B5(t=0,pi/2) { x = R2*cos(-t); y = H2*sin(-t); label = 5;}

```

```

func bord1 = B1(-Lg1/hh) + B2(-Lg2/hh) + B3(-Lg3/hh) + B4(-Lg4/hh)
+ B5(-Lg5/hh) ;

```

```

mesh Th=buildmesh(bord1);
Th = adaptmesh(Th, hh, IsMetric=1, nbvx=1000000);
plot(Th, wait=1, fill=1);

```

```

macro sign(x) ((x)<0?-1.:1.)//
macro ssqrt(x) (sign(x)*sqrt(abs(x)))//
fespace Vh (Th, P2);
func FF = ssqrt(1-y*y)*(R1+D0*y*y) - x ;
func F=FF(x,min(0.,y/L0)) + max(0.,y);
//Vh Fh = F; //F(x,min(0.,y/L0));

```

```

Vh Fh2; // n+1
Vh eta;

```

```

savemesh(Th,"Th.msh");

```

```

real[int] v=[0];

```

```

real alpha=0.01;
func G=alpha*(-(x^2)-(y-50)^2+40^2);
Vh Fh = G;
plot(Fh,viso=v, wait=1, cmm="Gh");
cout << "long Fh=" << int1d(Th,levelset=Fh)(1.) << endl;

```

```

// Yu, Sakai, Sethian, IFB 2003
real rhoF=1070 * 1e-12; // mg/(micro m^3)
real muF =3.34 * 1e-3 ; // mg/(micro m s)

```

```

real rhoG=1.225 * 1e-12; // mg/(micro m^3)
real muG =1.77625 * 1e-3; // mg/(micro m s)

```

```

//real gamma=0.032 * 1e6 ; // mg/(s^2)
real gamma=0.000001; // mg/(s^2)

```



```

real t;
real Vin=10; // micro m /s

func f1=0;
func f2=0;

func h1=0;
func h2=0;

real dt=1e-3; // delta t (s)
real delt=dt;

int NN=30; //

fespace Wh(Th,P1b);
fespace Qh(Th,P1);

Wh u1h,u2h; // u^{n+1}
Wh u1hp,u2hp; // u^n
Wh v1h,v2h;
Qh ph; // p^{n+1}
Qh qh;
Qh courbe;

Qh caractF ,caractG ;

caractF = (Fh(x,y) >= 0 ? 1 : 0 );
//plot(caractF ,wait=1,value=true , fill=true ,dim=3,cmm="caractF ");

caractG=1-caractF ;
//plot(caractG ,wait=1,value=true , fill=true ,dim=3,cmm="caractG ");

//plot(caractF ,caractG ,wait=1,value=true , fill=true ,dim=3,cmm="caractF caractG ");

courbe= (-((dy(Fh))^2)*dxx(Fh) + 2*dx(Fh)*dy(Fh)*dxy(Fh)-
((dx(Fh))^2)*dyy(Fh))/((dx(Fh)*dx(Fh) + dy(Fh)*dy(Fh))^2);
//plot(courbe ,wait=1,value=true , fill=true ,dim=3,cmm="courbe ");

problem Stokes(u1h,u2h,ph,v1h,v2h,qh)=
int2d(Th)( u1h *v1h*caractF*rhoF/dt + u2h *v2h*caractF*rhoF/dt )
-int2d(Th)( u1hp*v1h*caractF*rhoF/dt + u2hp*v2h*caractF*rhoF/dt )
+int2d(Th)(
2*muF*caractF*(

```

```

dx(u1h)*dx(v1h) + 0.5*( dy(u1h)+dx(u2h) )*( dy(v1h)+dx(v2h) ) + dy(u2h)*dy(v2h)
)
)
-int2d(Th)( (dx(v1h)+dy(v2h))*ph*caractF )
+int2d(Th)( u1h*v1h*caractG*rhoG/dt + u2h*v2h*caractG*rhoG/dt )
-int2d(Th)( u1hp*v1h*caractG*rhoG/dt + u2hp*v2h*caractG*rhoG/dt )
+int2d(Th)(
2*muG*caractG*(
dx(u1h)*dx(v1h) + 0.5*( dy(u1h)+dx(u2h) )*( dy(v1h)+dx(v2h) ) + dy(u2h)*dy(v2h)
)
)
-int2d(Th)( (dx(v1h)+dy(v2h))*ph*caractG )

+int1d(Th,levelset=Fh)( gamma*(dx(Fh)*v1h+ dy(Fh)*v2h)*
(-(dy(Fh))^2)*dxx(Fh) + 2*dx(Fh)*dy(Fh)*dxy(Fh)-
((dx(Fh))^2)*dyy(Fh))/((dx(Fh)*dx(Fh) + dy(Fh)*dy(Fh) )^2)
)

-int2d(Th)( ( dx(u1h)+dy(u2h) )*qh )

-int2d(Th)( f1*v1h + f2*v2h )
-int1d(Th,5)( h1*v1h + h2*v2h )
+on(1,u1h=0)
+on(2,u1h=0, u2h=Vin*(x+R1)*(x-R1)/(R1)^2 )
+on(3,4,u1h=0,u2h=0);

problem Deplacement(Fh2,eta)=
int2d(Th)( ( (u1h*dx(Fh2) + u2h*dy(Fh2))*dt)*
( eta+delt*( u1h*dx(eta)+u2h*dy(eta) ) ) )
+ int2d(Th)( Fh2 *( eta+delt*( u1h*dx(eta)+u2h*dy(eta)
) ) )
- int2d(Th)( Fh *( eta+delt*( u1h*dx(eta)+u2h*dy(eta) ) )
) )

//+on(2,Fh2=0.1-0.1*x);
+on(2,Fh2=Fh);

u1hp=0;
u2hp=0;

for(int n=0;n<NN;n++){

t=(n+1)*dt;

```

```

cout << "t= " << t << endl;

Stokes;
//plot([u1h,u2h],wait=0,value=true,cmm="n="+n);
//plot(ph,wait=0,dim=3,value=true,fill=true,cmm="n="+n);

u1hp=u1h;
u2hp=u2h;

cout << "courbure/ (..)1/2 =" << int1d(Th,levelset=Fh)
( -(dy(Fh))^2)*dxx(Fh) +
2*dx(Fh)*dy(Fh)*dxy(Fh)- ((dx(Fh))^2)*dyy(Fh))/
((dx(Fh)*dx(Fh) + dy(Fh)*dy(Fh))^2))
<< endl;

Deplacement;

Fh = Fh2;

//Fh= Fh+ dt*(u1h*dx(Fh)+u2h*dy(Fh));

plot(Fh,viso=v,wait=0,value=true,cmm="n="+n);

caractF = (Fh(x,y) >= 0 ? 1 : 0 );

caractG=1-caractF;

};

```